

GLPK Case Study 1 - 60016 Operations Research

The case study teaches how to use GLPK to solve a basic resource allocation problem inspired to a realistic situation and explains how to read the `glpsol` output files. Consider a company producing three brands of fertilisers:

SulphurGrower (S)
EssarGrower (E)
NitroGrower (N).

The raw materials required to produce each of these are:

SulphurGrower	EssarGrower	NitroGrower
15% Sulphur	10% Sulphur	5% Sulphur
5% Nitrate	10% Nitrate	10% Nitrate
10% Potash	10% Potash	15% Potash

During any given period the company can obtain up to 1000 kg of Sulphur, 1600 kg Nitrate, and 2500 kg Potash. During the same period the company has contracts to supply at least 1000 kg of SulphurGrower (S), and 500 kg of EssarGrower (E). There is no upper limit on the volumes of fertilisers that the company can sell. The profits for each fertiliser are:

SulphurGrower £3/kg
EssarGrower £5/kg
NitroGrower £5/kg

1. Write the LP for maximising the profit.
2. Using GMPL, write a corresponding LP model for GLPK.
3. Using `glpsol`, determine the optimal solution of this model.
4. If the optimal solution is implemented by the company, what raw materials will be the limiting factors for the production?

Solution 1. Denote the variables in the problem by:

x_S Production of SulphurGrower in kg,
 x_E Production of EssarGrower in kg,
 x_N Production of NitroGrower in kg.

The LP for maximising the profit is then:

$$\begin{aligned} \max \quad & y = 3x_S + 5x_E + 5x_N \\ \text{subject to:} \\ 0.15x_S + 0.10x_E + 0.05x_N & \leq 1000 \\ 0.05x_S + 0.10x_E + 0.10x_N & \leq 1600 \\ 0.10x_S + 0.10x_E + 0.15x_N & \leq 2500 \\ x_S \geq 1000, x_E \geq 500, x_N & \geq 0. \end{aligned}$$

2. Using GMPL, we can specify the above LP as follows:

Listing 1: growers.mod

```

set Growers;
set RawMaterials;
param Profit {g in Growers};
param RawSupply {r in RawMaterials};
param Contracts {g in Growers};
param Ratio {r in RawMaterials, g in Growers}; /* percentage of r within g */
                                     Row First      Column Second
var x {g in Growers}, >= 0; /* production of grower g */

maximize y : sum {g in Growers} Profit[g] * x[g];
                                     Iterate by row: For each raw material, the used
                                     amount has to be smaller than the Raw Supply
s.t.
c1 {r in RawMaterials} : sum {g in Growers} Ratio[r,g] * x[g] <= RawSupply[r];
c2 {g in Growers} : x[g] >= Contracts[g];

data;

set RawMaterials := Sulphur Nitrate Potash ;
set Growers := SulphurGrower EssarGrower NitroGrower ;

param Profit :=
SulphurGrower 3
EssarGrower 5
NitroGrower 5 ;
                                     As defined in previous section: {g in Growers}, the
                                     list has to use Growers components as index

param RawSupply :=
Sulphur 1000
Nitrate 1600
Potash 2500 ;

param Contracts :=
SulphurGrower 1000
EssarGrower 500
NitroGrower 0 ;

param Ratio : SulphurGrower EssarGrower NitroGrower :=
Sulphur      0.15      0.10      0.05
Nitrate      0.05      0.10      0.10
Potash       0.10      0.10      0.15 ;

end;

```

3. Saving the GMP model in `growers.mod`, we can then run `glpsol -m growers.mod -o growers.out` and examine the output file:

Listing 2: `growers.out`

```

Problem:    growers
Rows:       7
Columns:    3
Non-zeros:  15
Status:     OPTIMAL
Objective:  y = 80700 (MAXimum)

```

Activity = Optimal Values for Decision Variables

No.	Row name	St	Activity	Lower bound	Upper bound	Marginal
1	y	B	80700			
2	c1[Sulphur]	NU	1000		1000	4
3	c1[Nitrate]	NU	1600		1600	48
4	c1[Potash]	B	2410		2500	
5	c2[SulphurGrower]	B	1400	1000		
6	c2[EssarGrower]	NL	500	500		-0.2
7	c2[NitroGrower]	B	14800	-0		

With 1 more unit of resources, how much the objective function (i.e., profit) will increase

No.	Column name	St	Activity	Lower bound	Upper bound	Marginal
1	x[SulphurGrower]	B	1400	0		
2	x[EssarGrower]	B	500	0		
3	x[NitroGrower]	B	14800	0		

... [Ignore the rest]

The output file shows two tables:

- a *rows table* that shows in the **Activity** column the value of the objective (row y) and how the optimal solution satisfies the constraints (rows c1[Sulphur] to c2[NitroGrower])
- a *columns table* (rows x[SulphurGrower] to x[NitroGrower]), which shows in the **Activity** column the values of the decision variables in the optimal solution.

According to `growers.out`, the optimal solution is $x^* = (x_S^*, x_E^*, x_N^*) = (1400, 500, 14800)$ and the objective function has an optimal value $y^* = 80700$ (Objective row).

4. Inspecting `growers.out`, we see that the left-hand sides of c1[Sulphur] and c1[Nitrate] are limited by the upper bound, as visible in the **Activity** column for the *rows table*. If the plan is implemented, the company will use 1000 kg of sulphur and 1600 kg of nitrate, and 2410 kg of potash. Therefore, sulphur and nitrate will be the scarce materials that limit the production. Potash is not a limiting factor since up to 2500 kg can be obtained, but we require just 2410 kg.